
Lab 5: A Nonlinear System

This Maple lab is based in part on earlier versions prepared by Professors R. Falk and R.
Bumby of the Rutgers Mathematics department.

Introduction. In this lab we use Maple to study the phase plane of an autonomous nonlinear
system of two differential equations, i.e., a system of the form

x′(t) = F (x, y), y′(t) = G(x, y). (1)

The distinguishing feature of an autonomous system is that the expressions defining the functions
F and G do not contain the independent variable t. This allows many properties of the solutions
to be studied using the curves, called trajectories, that show the paths in the xy plane followed by
the solutions. (It is an easy exercise to show that if an initial condition is on a trajectory, then
the whole solution follows that trajectory). The Maple command DEplot may be used to draw
trajectories and direction fields for such systems.

Please obtain the seed file from the web page and save it in your directory on eden, then
prepare the Maple lab according to the instructions and hints in the introduction to Lab 0. Turn in
only the printout of your Maple worksheet, after removing any extraneous material and any errors
you have made.

0. Setup. As usual, the seed file begins with commands which load the required Maple packages:
with(plots):, with(DEtools):, and with(LinearAlgebra):.

1. An autonomous nonlinear system: a. The equation and its critical points.

We will study the nonlinear system

x′ = −x2 + xy + y + 1 = (x+ 1)(y − x+ 1)

y′ = −x2 − xy + 5x+ 4y − 4 = (x+ y − 1)(−x+ 4)
(2)

The equilibrium solutions are [x = −1, y = 2], [x = 1, y = 0], and [x = 4, y = 3]. Maple can obtain
these by using the instructions

F:=-(x+1)*(x-y-1);

G:=(x+y-1)*(-x+4);

critpts:=solve(F,G,x,y);

which are included in the seed file. Also included are instructions to define the differential equations:

dex:=diff(x(t),t)=eval(F,{x=x(t),y=y(t)});

dey:=diff(y(t),t)=eval(G,{x=x(t),y=y(t)});

For later convenience we have defined F and G to depend on the variables x and y, but in the
differential equations we must write these variables as x(t) and y(t); the eval command makes this
substitution.

b. The direction field and nullclines. The next instructions establish an appropriate range
for the independent variable and a useful viewing window, and construct a plot of the direction
field of this system. The plot consists of small arrows pointing the way of the trajectories in the
square −5 ≤ x ≤ 5, −5 ≤ y ≤ 5.

trange := -5..5: window:=x=-5..5,y=-5..5:

df:=DEplot([dex,dey],[x(t),y(t)], trange, window,color=GREEN):

(Note the colon at the end of the instruction defining the plot, which suppresses output of the plot
structure). The color option is used to give a better view of the nullclines and solution curves to
be added later. If you like, you can test different values of the option dirfield, as in Lab 2, and
customize the value for the plots in this project. A value should be chosen that allows individual
arrows to be identified while including a sufficient number of arrows. To show this plot, you need
only enter its name df1; after defining it. The result should not be left in the worksheet, since we
will include this as part of a display command in the next subsection.

Nullclines. The points where the direction field is horizontal (characterized by dy/dt = 0) or vertical
(characterized by dx/dt = 0) form curves called nullclines. In many cases, these curves provide
useful information about the behavior of trajectories without the excessive detail of a direction
field.

For the system (2), the factored form of the expressions for dx/dt and dy/dt allows a simple
description of the nullclines. In particular, since x′ = (x+ 1)(y − x+ 1), the slope field is vertical
along the curves x = −1 and y = x − 1. Since x = −1 is a vertical line, it is best to plot these
curves parametrically; such a plot can be constructed using

dh:=plot([[-1,t,t=trange],[t,t-1,t=trange]],window,color=CORAL):

Construct a similar instruction to produce a plot dh showing where the slope field is horizontal; to
be definite, use color=BLUE. Now obtain a display (including a title) combining the slope field
and both sets of nullclines, using the following instruction in the seed file:

display(df,dh,dv,title="Slope field and nullclines: Nonlinear equation");

The equilibrium points should be seen as points lying on one nullcline of each color. Note that
the nullclines usually cut across the arrows in the slope field since they are not trajectories of the
system; the nullcline x = −1 is exceptional.

Discussion: For each of the three equilibrium points found in part a, give explicitly the equations
of the vertical and the horizontal nullclines which cross at that point.

c. Trajectories. To study the behavior of the system and the nature and stability of equilibrium
points, it is also useful to construct a plot of the phase plane which shows several trajectories as
well as the direction field. In Lab 4, we did this by finding the exact solution to the equation. Since
this is in general not possible for nonlinear equations, we use here the numerical methods that are
part of the DEplot command. To produce each trajectory one specifies an initial condition—i.e., a
point in the phase plane that is to correspond to t = 0—and asks Maple to produce the trajectory
through that point; to get a significant part of the full trajectory one solves over a range of the
independent variable, t, which includes both positive and negative values.

To obtain good choices of initial conditions, leading to a set of trajectories which illustrate the
important features of the phase plane, requires some trial and error. The seed file contains a list
of six initial conditions, together with a command to produce a plot of the phase plane containing
the direction field and the corresponding six trajectories:

inits:=[[x(0)=3,y(0)=4],[x(0)=2,y(0)=-1],[x(0)=5,y(0)=0],

[x(0)=1,y(0)=0.2],[x(0)=-.5,y(0)=.5],[x(0)=-2,y(0)=3]]:

DEplot([dex,dey],[x(t),y(t)],t=trange,inits,window,color=GREEN,

linecolor=[RED,BLUE,BROWN,PLUM,CORAL,BLACK],thickness=2,stepsize=0.005,

title=‘‘Trajectories: Nonlinear equation’’);

The choice of stepsize here seems to work well, but you can experiment with other possible choices.
Include the resulting plot in your final worksheet.

Discussion: By examining the phase plane plot, you should be able to decide the type of each of
the three critical points: stable or unstable node, saddle point, or stable or unstable spiral. Give
your conclusions for each point in the discussion section.

Discussion: In the second discussion section you should describe any interesting properties of each
of the six trajectories shown. In particular, for each one, discuss whether or not that trajectory
appears to approach a critical point—and if so, which one—as t → ∞ and as t → −∞. In some
cases you will not be able to determine this from your graph, and it is perfectly acceptable to say
so. In your discussion you should identify the various trajectories by color.

d. Linearization. The type and stability of the critical points can be determined by examining
the eigenvalues of the corresponding linear system. The matrix of the linearization of the system
(1) at the critical point (xc, yc) is

A :=

(
Fx(xc, yc) Fy(xc, yc)
Gx(xc, yc) Gy(xc, yc)

)
.

Maple can find this matrix in a two step process, first constructing the matrix of partial derivatives
(which need be done only once), then evaluating it at the critical point (done separately for each
point). Instructions for finding A for general x and y and then finding A1, its value at the critical
point (−1, 2), and the eigenvalues of A1, are in the seed file:

A := Matrix([[diff(F,x),diff(F,y)],[diff(G,x),diff(G,y)]]);

A1 := eval(A,{x=-1,y=2});

Eigenvalues(A1);

Use Maple to find the matrices A2 and A3, and their eigenvalues, corresponding to the critical
points on the x-axis and in the first quadrant, respectively.

Discussion: In a discussion section, for each critical point, describe what the eigenvalues you found
imply about the type of that point, and compare the answer with the one you found in part c.
(The answers should agree!)

The phase plane near the critical points. In sections e, f, and g we sketch the phase plane
for the true system, and separately for its linearization, near each of the three critical points. We
use a square window, 0.4 units by 0.4 units, centered at the critical point, and as initial conditions
the four points (xc ± 0.1, yc) and (xc, yc ± 0.1).

Discussion: In each section you should produce two plots. Include also a discussion comparing the
plots for the true and linearized systems: what similarities do you see? What differences?

e. The phase plane near (−1, 2). For this case the seed file contains the commands needed
to produce the two plots. The commands needed to graph the phase plane of the nonlinear system
near (−1, 2) are

trange1 := -3..3: window1 := x=-1.2..-0.8,y=1.8..2.2:

inits1:=[[x(0)=-1,y(0)=2.1],[x(0)=-1,y(0)=1.9],[x(0)=-1.1,y(0)=2],

[x(0)=-0.9,y(0)=2]]:

DEplot([dex,dey],[x(t),y(t)],t=trange1,inits1, window1,color=GREEN,

linecolor=[RED,BLUE,BROWN,BLACK],thickness=2,stepsize=0.002,

title=‘‘Phase plane near (-1,2): nonlinear system’’);

For the linearized system the commands are

F1:=4*u;

G1:=5*(u+v);

dex1:=diff(x(t),t)=eval(F1,u=x(t)+1,v=y(t)-2);

dey1:=diff(y(t),t)=eval(G1,u=x(t)+1,v=y(t)-2);

DEplot([dex1,dey1],[x(t),y(t)],t=trange1,inits1, window1,color=GREEN,

linecolor=[RED,BLUE,BROWN,BLACK,thickness=2,stepsize=0.002,

title=‘‘Phase plane near (-1,2): linearized system’’);

Note that here:

(i) The functions F1 and G1 are obtained from the matrix A1 found in part d.
(ii) In defining the linear differential equations we make the substitutions u = x− xc = x+ 1 and

v = y − yc = y − 2, so that the resulting phase portrait is centered (in the xy plane) at the
point (−1, 2). This provides for easy comparison with the nonlinear case.

Discussion: In your discussion, explain why this nonlinear system and linear system share a
pair of straight line trajectories.

f. The phase plane near (1, 0). Modify the above commands to produce plots of the phase
plane for the nonlinear and linear systems near (1, 0). Include a fifth trajectory, colored CORAL,
with initial condition [x(0) = 0.8, y(0) = 0.1]. This initial point is of the form (1, 0) + 0.1(ξ1, ξ2),
where ξ = (−2, 1) is one of the eigenvectors for the linear problem (found using the command
Eigenvectors(A2)). The corresponding trajectory is a straight line in the linear system, but not
in the nonlinear system.

Discussion: In your discussion, explain why this (coral) trajectory looks so different in the two
plots, while the other trajectories are similar.

g. The phase plane near (4, 3). Modify the above commands to produce plots of the phase plane
for the nonlinear and linear systems near (4, 3), and again compare the plots in your Discussion.

End of Lab 5

